

Class Jobs

The class jobs problem is a jobs planning problem where the satisfaction of the children is maximized by assigning them to the best jobs.

Problem

- Maximize the aggregated wishes of the children for the jobs.

Sets

- $c \in Children = \{1, 2, 3, 4, 5\}$
- $j \in Jobs = \{1, 2, 3, 4, 5\}$

Parameters

- $Wish_{j,c}$: How much child c wish job j , 1 for worst, 5 for best

Decision variables

- Assignment variable of, if 1 child c performs job j $x_{j,c} \geq 0$.

Model

Objective:

- Maximize the wish fulfillment:

$$\sum_{j,c} Wish_{j,c} \cdot x_{j,c}$$

Constraints:

- Each child c should be assigned to one job:

$$\sum_j x_{j,c} = 1 \quad \forall c$$

- Each job j should be assigned to one child:

$$\sum_c x_{j,c} = 1 \quad \forall j$$

The above model is a simple assignment model. Notice that because of the structure of the model, the variables $x_{j,c}$ will only take the values 0 or 1.

The full model in Julia/JuMP, available with the name

`ClassJobs.jl`

from the book web-site, is given below:

```

*****
# Class Jobs assignment, LP
using JuMP
using HiGHS
*****

# Data
Children=[1 2 3 4 5]
C=length(Children)
Jobs=[1 2 3 4 5]
J=length(Jobs)
Wish=[
1 3 2 5 5;
5 2 1 1 2;
1 5 1 1 1;
4 5 4 4 4;
3 5 3 5 3]
*****

# Model
CJ = Model(HiGHS.Optimizer)

@variable(CJ,x[j=1:J,c=1:C]>=0)

# maximize aggregated Wish

```

```

@objective(CJ, Max, sum( Wish[j,c]*x[j,c] for j=1:J,c=1:C ) )

# One job pr. child
@constraint(CJ, [c=1:C],
            sum( x[j,c] for j=1:J) == 1
            )

# One child pr. job
@constraint(CJ, [j=1:J],
            sum( x[j,c] for c=1:C) == 1
            )

#####

#####
# Solve
solution = optimize!(CJ)
println("Termination status: $(termination_status(CJ))")
#####

#####
if termination_status(CJ) == MOI.OPTIMAL
    println("Optimal objective value: $(objective_value(CJ))")
    for c=1:C
        for j=1:J
            if value(x[j,c])>0.999
                println("Child: ", c, " Doing job: ", j)
            end
        end
    end
else
    println("No optimal solution available")
end
#####

```